Wzory i programy

Andrzej WALAT



Kilka lat temu kolega – Jan Dunin Borkowski – zapytał mnie: Czy licealista ma szanse poradzić sobie z następującym zadaniem?

W Polsce jest 16 *miast wojewódzkich. Co tydzień losujemy jedno z miast. Co jest bardziej prawdopodobne: zdarzenie* A*, że po* 16 *takich losowaniach liczba miast co najmniej raz wylosowanych będzie niewiększa niż* 8 *(połowa wszystkich miast), czy zdarzenie przeciwne* B*, że wylosujemy ich więcej niż* 8*?*

Odpowiedziałem, że owszem – ma. A w jaki sposób? Można zacząć od analogicznego, ale prostszego problemu – przyjąć, że liczba miast m i liczba losowań l jest mniejsza, np. $m = l = 4$. To zadanie nietrudno rozwiązać posługując się grafem przedstawionym na rysunku. W polach wiersza n (wiersze numerujemy od góry zaczynając od 1) są liczby mówiące, ile różnych miast można wylosować po n losowaniach. Np. po trzech losowaniach liczba wylosowanych miast może być równa 1, 2 albo 3. Liczba na strzałce od pola i w wierszu n do pola i (albo $i+1$) w wierszu $n+1$ to prawdopodobieństwo zdarzenia, że liczba miast wylosowanych po $n+1$ losowaniach będzie równa i (albo $i+1$), jeśli w wyniku n losowań wylosowano i różnych miast.

Stosując tzw. prawa dodawań i mnożeń można obliczyć:

$$P(A) = \frac{1}{4}\cdot\frac{1}{4}\cdot\frac{1}{4} + \frac{1}{4}\cdot\frac{1}{4}\cdot\frac{3}{4} + \frac{1}{4}\cdot\frac{3}{4}\cdot\frac{2}{4} + \frac{3}{4}\cdot\frac{2}{4}\cdot\frac{2}{4} = \frac{11}{32} \quad \text{oraz} \quad P(B) = 1 - P(A) = \frac{21}{32}.$$

Zdarzenie, że wylosujemy więcej niż połowę miast jest prawie dwa razy bardziej prawdopodobne. Cierpliwy Czytelnik może obliczyć, że również w przypadku 6 miast i 6 losowań zdarzenie B wygrywa z A i to w jeszcze większym stosunku. Możemy przypuszczać, że również w przypadku 16 miast i 16 losowań rozstrzygnięcie będzie podobne. Jak możemy to sprawdzić?

Niech $P(m, l, r)$ oznacza prawdopodobieństwo zdarzenia, że w wyniku l losowań jednego z m miast wylosujemy r różnych miast. Zakładamy, że m, l i r to dowolne dodatnie liczby całkowite spełniające następujące ograniczenia: $l \leqslant m$ oraz $r \leqslant \min(l, m)$. Rysunek podpowiada następującą zależność rekurencyjną:

1. dla $r = l = 1$, $P(m, l, r) = 1$;
2. dla $r = 1$ i $l > 1$, $P(m, l, r) = P(m, l-1, r)/m$;
3. dla $r = l$ i $l > 1$, $P(m, l, r) = P(m, l-1, r-1)(m+1-l)/m$;
4. w pozostałych przypadkach

$$P(m, l, r) = P(m, l-1, r-1)(m+1-r)/m + P(m, l-1, r)r/m.$$

Zapisem tych wzorów w Logo jest następująca procedura.

```
oto p :m :l :r
jeśli i :r = 1 :l = 1 [wynik 1]
jeśli :r = 1[wynik (p :m :l - 1 :r) / :m]
jeśli :r = :l[wynik (p :m :l - 1 :r - 1) * (:m + 1 - :l) / :m]
wynik ((p :m :l - 1 :r - 1) * (:m + 1 - :r) + (p :m :l - 1 :r) * :r) / :m
już
```

Przyda się jeszcze funkcja `pA` dająca w wyniku prawdopodobieństwo zdarzenia A dla dowolnej parzystej liczby miast $m = 2n$.

```
oto pA :n
niech "w 0
powtórz :n [przyp "w :w + p 2 * :n 2 * :n npw]
wynik :w
już
```

Możemy teraz obliczyć prawdopodobieństwo zdarzenia A dla dziesięciu kolejnych parzystych wartości m od 1 do 20. Po napisaniu polecenia

```
powtórz 10 [ (ps "|m = | 2 * npw "| p(A) = | pA npw)]
```

komputer wypisał:

Rozwiązanie zadania F 729.
Aby dostroić się do danej stacji radiowej, musimy zmienić parametry układu drgającego tak, aby częstotliwość własna drgań w tym układzie była równa częstotliwości fali radiowej, która odpowiada danej stacji. Zatem aby przejść do odbioru fal dłuższych, należy zwiększyć okres drgań własnych obwodu drgającego ($\lambda = cT = 2\pi c\sqrt{LC}$), czyli tym samym zwiększyć pojemność kondensatora w obwodzie – jest to prostsze od manipulowania indukcyjnością cewki. Pojemność kondensatora płaskiego jest odwrotnie proporcjonalna do odległości między okładkami, zatem aby zwiększyć jego pojemność, należy zmniejszyć odległość między płytkami.

```
m = 2 p(A) = 0.5
m = 4 p(A) = 0.34375
m = 6 p(A) = 0.251543
m = 8 p(A) = 0.190102
m = 10 p(A) = 0.146461
m = 12 p(A) = 0.114296
m = 14 p(A) = 0.0900198
m = 16 p(A) = 0.0713935
m = 18 p(A) = 0.0569297
m = 20 p(A) = 0.0455953
```

Dla $m = 16$ – jak w oryginalnym sformułowaniu zadania o losowaniu miast – $P(A) < 0.1$. Na tym moglibyśmy zakończyć badanie problemu, gdyby nie jedno ale. Czytelnik, który przepisał podane wyżej definicje funkcji `p` oraz `pA` i wywołał polecenie powodujące wypisanie kolejnych dziesięciu wartości $P(A)$ dla m od 2 do 20, musiał zauważyć, że komputer wypisał szybko pierwszych 8 wyników, ale na wydrukowanie kolejnych wartości dla $m = 18$ oraz szczególnie 20, trzeba już było trochę poczekać. Obliczenie $P(A)$ dla każdej kolejnej parzystej wartości m trwa w przybliżeniu czterokrotnie dłużej. Gdyby w Polsce było 50 miast wojewódzkich, to nigdy nie doczekalibyśmy się odpowiedniego wyniku funkcji `pA`.

Alternatywne rozwiązanie problemu

Z każdym wierszem grafu na rysunku, a także podobnego bardziej złożonego grafu, można związać pewną zmienną losową – funkcję, która przyjmuje losowo wartości liczbowe z określonymi prawdopodobieństwami. Na przykład z wierszem numer 2 wiążemy zmienną R_2 – liczbę różnych wylosowanych miast po dwóch losowaniach. Zmienna R_2 przyjmuje wartości [1 2] z prawdopodobieństwami $[\frac{1}{4}\ \frac{3}{4}]$. Kolejna zmienna R_3 przyjmuje wartości [1 2 3] z prawdopodobieństwami $[\frac{1}{16}\ \frac{9}{16}\ \frac{6}{16}]$. Oczywiście zmienna R_1 przyjmuje tylko jedną wartość 1 z prawdopodobieństwem 1. W tym rozwiązaniu będziemy operowali nie na pojedynczych liczbach – wartościach prawdopodobieństw, ale na rozkładach prawdopodobieństw reprezentowanych w postaci list liczb.

Zdefiniujemy funkcję, która dla dowolnej liczby miast m oraz liczby losowań $l \leqslant m$ wyznacza rozkład odpowiedniej zmiennej losowej R_l.

```
oto rozkład :m :l
jeśli :l = 1 [wynik [1]]
wy kolejny rozkład :m :l - 1
już
```

Treść tej definicji jest chyba zrozumiała. Dla $l = 1$ wynikiem jest rozkład zdefiniowany przez jednoelementową listę [1]. Dla większych l wynikiem jest kolejny rozkład po rozkładzie $l - 1$. Ale jak zdefiniować funkcję, która wyznacza kolejny rozkład, mając dany jakiś aktualny rozkład?

```
oto kolejny :ar
niech "w1 ilep :ar / :m
niech "w2 :ar - :w1
wynik (nap 0 :w2) + nak 0 :w1
już
```

Tajemnicze `ilep` – to skrótowa nazwa operacji *iloczyn element razy pozycja*, która mnoży każdy element listy liczb przez numer jego pozycji na liście. Można ją zdefiniować na przykład tak:

```
oto ilep :ll
jeśli puste? :ll [wy []]
wynik nak (ost :ll) * (długość :ll) ilep bo :ll
już
```

Wynikiem operacji `ilep` dla danej listy kolejnych liczb naturalnych [1 2 3 4 ...] będzie lista kolejnych liczb kwadratowych [1 4 9 16 ...].

Kolejny rozkład wyznaczamy w następujący sposób. Tworzymy pomocniczy wektor (listę liczb) w_1, dzieląc przez m wynik operacji `ilep` na danym aktualnym rozkładzie. Tworzymy wektor w_2, odejmując w_1 od aktualnego rozkładu. Sumujemy w_2 z dopisanym na końcu zerem oraz w_1 z zerem dopisanym na początku i to jest wynik – kolejny rozkład.

Funkcję, która daje w wyniku prawdopodobieństwo zdarzenia A dla danej parzystej liczby miast $m = 2n$, można teraz zdefiniować w następujący sposób.

```
oto prA :n
wynik sumaWycinka 1 :n rozkład 2 * :n 2 * :n
już
```

```
oto sumaWycinka :p :k :listaLiczb
jeśli :p > :k [wynik 0]
wynik (element :p :listaLiczb) + sumaWycinka :p +
1 :k :listaLiczb
już
```

Po napisaniu polecenia

```
powtórz 25 [ (ps "|m = | 2 * npw "| p(A) = | prA npw)]
```

na ekranie w ułamku sekundy pojawi się 25 kolejnych wartości $P(A)$ dla kolejnych parzystych wartości m od 2 do 50. Dla $m = 50$ otrzymujemy $P(A) = 0.00209116$.

Na koniec zagadka dla dociekliwego Czytelnika: na czym polega tajemnica tak znacznego przyspieszenia obliczeń?